

Title	パーソナル・コンピュータPC8801を用いたインテリジェント・ターミナル
Author(s)	藤井, 博
Citation	大阪大学大型計算機センターニュース. 1982, 46, p. 83-94
Version Type	VoR
URL	<a href="https://hdl.handle.net/11094/65538">https://hdl.handle.net/11094/65538</a>
rights	
Note	

*Osaka University Knowledge Archive : OUKA*

<https://ir.library.osaka-u.ac.jp/>

Osaka University

# パーソナル・コンピュータ PC8801を 用いたインテリジェント・ターミナル

大型計算機センター

共同利用掛 藤 井 博

## 1. はじめに

日進月歩の技術革新に伴い、コンピュータも小型化し、それに反比例して処理能力が向上してきている。それに伴い利用者の処理傾向もセンターに足を運ばなくても、研究室に居ながら利用できるTSSが主流を占める様に変化してきている。一方、世の中は数年前から始まったマイコン・ブームで最近では、ミニコンに優るとも劣らない様な機種が低価格で手に入る様になり、これを端末として使用している利用者も少なくないと思う。

特に最近では、BASICが非常に強力になり（そのためBASICインタープリンタがバカデカくなったが）数十行のステートメントでインテリジェント・ターミナルのプログラムが書ける様になったのでここに紹介することにした。

## 2. 機器構成と機能

機器構成(表1)は図1のように構成した。PC8801には、TERMコマンドが用意されているが、当センターで試したところ、表示が正常におこなわれず、まったく使いものにならなかった。せめて同じメーカーのホストとはすんなり接続できる様にしておいてほしいものである。<sup>(注1)</sup>

インテリジェント・ターミナルの機能としては、次のようなものを考慮した。

- (1) キャラクターディスプレイ・ターミナルとして動作すること。
- (2) スクリーンエディッタが使用できること。
- (3) 必要な時、ハードコピーがとれること。

パーソナル・コンピュータ	NEC	PC8801
カラーディスプレイ	NEC	PC8853
8インチフロッピーディスク	NEC	PC8881
5インチフロッピーディスク	NEC	PC8031-2W
プリンター	NEC	PC8023-C
音響カプラー		

表1. 構成機器

- (4) ホストから送信されてくるデータを、フロッピーディスクに書き込むこと。
- (5) オフラインであらかじめ作ったフロッピーディスクの内容をホストのファイルに転送できること。
- (6) 640×400ドットのグラフィックディスプレイとして動作すること。

(注1) N<sub>88</sub>-BASIC Version 1.1 では、TERMコマンドが修正されACOSにも接続できる様です。

非常に欲張って6つ機能をすべて動作するプログラムを、と考えて見たけれど、最後のグラフィックディスプレイ機能だけは、BASICインタープリンタの悲しさが処理スピードが追いつかないのでこの機能を断念した。

さて、回線スピードであるが、300 BPS以上のスピードで通信しようと思うと、どうしても機械語でプログラムを書かなければならない。ところがこのメーカーのパーソナル・コンピュータでも同じであるが、I/O ポートアドレスとか ROM内ルーチンのアドレス

等は公開されていず利用者の手で解析するしか方法がないので、とりあえずBASICでプログラムを書いた。そのため回線スピードは、300BPSが限度と思われる。




図1. システム構成

### 3. プログラムの構成




図2. フローチャート

PC8801には、10個のファンクションキーがあり各キーに最大15文字まで定義でき、また各ファンクションキーを割込みキーとしても使用できる。

今回のプログラムは、このファンクションキー個々に機能を定義し、割込み動作させることにより、図2のシステムフローチャートの様な非常にスッキリしたプログラムとなった。

まず、スクリーンの大きさ等の設定、RS232Cのオープン、各ファンクションキーの割込み処理先の定義を行ない、そしてメインプログラムに入る。

メインプログラムでは、RS232Cからの受信、キーボードからの入力、ディスプレイ上への表示、RS232Cへの送信を行なう。フロッピーディスク等の書き込み処理は、ファンクシ

キーからの割込みにより、サブルーチンで処理する。


以下にプログラムを順に説明する。

```

10 'Terminal program for NEC PC8801 by H.FUJII
20 'Initial Set
30 WIDTH 80,25:CONSOLE 0,25,0,1:COLOR 4:PRINT CHR#(&HC):'Display set
40 DEFINT A-Z:CR#=CHR#(&HD):LOGING=0:DEL#=CHR#(&H7F):LF#=CHR#(&HA):GS=0
50 LISTING=0:GDP=0:GS#=CHR#(&H1D):ESC#=CHR#(&H1B):CAN#=CHR#(&H18):FF#=CHR#(&HC):
ETB#=CHR#(&H17):E0#=CHR#(&HE0):H20#=CHR#(&H20):H30#=CHR#(&H30):H40#=CHR#(&H40):U
S#=CHR#(&H1F):H16#=CHR#(&H16)
60 'RS232C open
70 OPEN "com:E71 S" AS #1:'RS232C initial set
80 'Function key set
90 HELP ON:KEY ON:STOP ON:'Activate all function keys and help key
100 ON HELP GOSUB 890:ON STOP GOSUB 650:ON ERROR GOTO 400
110 ON KEY GOSUB 490,570,610,330,950,680,730,760,920,530

```

行番号	説明
10	コメント
20	コメント
30	ディスプレイを1行80文字25行カラーモードにセットし画面をクリア。
40	} AからZで始まる変数を整数型として定義し、定数を定義。
50	
60	コメント
70	RS232Cをファイルコード#1ランダムファイルとしてオープン。 ACOSと接続するには、RS232Cを次の様に定義。



80	コメント
90	<span style="border: 1px solid black; padding: 2px;">HELP</span> キー, <span style="border: 1px solid black; padding: 2px;">ファンクション</span> キー, <span style="border: 1px solid black; padding: 2px;">STOP</span> キーを割込み処理として定義。
100	<span style="border: 1px solid black; padding: 2px;">HELP</span> キー押下, <span style="border: 1px solid black; padding: 2px;">STOP</span> キー押下, エラー発生時の処理先を定義。
110	ファンクションキー f・1 ~ f・10 押下時の処理先を定義。

```

120 'Main program
130 K#=INKEY#:IF K#="" THEN 230
140 'Send routine
150 IF K#=CHR#(&H1D) THEN GOSUB 300:GOTO 130
160 IF K#=CHR#(&H0) THEN GOSUB 300:GOTO 130
170 IF K#=CHR#(&H1C) THEN 130
180 IF K#=CHR#(&H1E) THEN 130
190 IF K#=CHR#(&H1F) THEN 130
200 COLOR 6:PRINT K#;:COLOR 4:PRINT #1,K#;
210 IF LISTING THEN LPRINT K#;
220 'Receive routine
230 IF LOC(1)=0 THEN 130
240 R#=INPUT$(1,#1)
250 IF R#=DEL$ THEN 230
260 PRINT R#;
270 IF LOGING THEN PRINT #2,R#;
280 IF LISTING THEN LPRINT R#;
290 GOTO 130

```

行番号	説 明
120	コメント
130	キーボード入力のチェック。
140	コメント(メインループ内のコメントなので、出来れば無い方がよい。)
150	<input type="checkbox"/> ← キー入力なら、バックスペース処理へ。
160	<input type="checkbox"/> CTRL + <input type="checkbox"/> H キー入力なら、バックスペース処理へ。
170	<input type="checkbox"/> → キー入力を無効にする。
180	<input type="checkbox"/> ↑ キー入力を無効にする。
190	<input type="checkbox"/> ↓ キー入力を無効にする。
200	キーボード入力文字の画面表示と、ホストへの送出。
210	ハードコピー指定があれば、プリント出力。
220	コメント(メインループ内のコメントなので、出来れば無い方がよい。)
230	ホストからの受信チェック。
240	受信バッファから1文字入力。
250	DEL(16進7F)コードの削除。
260	受信文字の画面表示。
270	フロッピーディスク書込み指定があれば、フロッピーディスク出力。
280	ハードコピー指定があれば、プリント出力。
290	メインループの先頭へ戻る。

```

300 'Back space routine
310 PRINT #1,CHR#(&H8);:LOCATE POS(0)-1,CSRLIN
320 PRINT " ";:LOCATE POS(0)-1,CSRLIN:RETURN
330 'edit mode routine
340 Y=CSRLIN:X=POS(0)
350 COLOR 6:LINE INPUT K#:COLOR 4
360 PRINT #1,K#;CR#;
370 LOCATE X,Y
380 RETURN
390 'Error routine
400 COLOR 2
410 IF ERR=23 THEN PRINT "Line buffer overflow":GOTO 470
420 IF ERR=62 OR ERR=56 THEN PRINT "Bad file name":GOTO 470
430 IF ERR=60 THEN PRINT "Disk space full":GOTO 470
440 IF ERR=53 THEN PRINT "File not found":GOTO 470
450 IF ERR=5 THEN PRINT "Back Space over":GOTO 470
460 ON ERROR GOTO 0
470 COLOR 4:RESUME 480
480 RETURN

```

行番号	説 明
300	コメント
310	ホストにBSコードを送出し、カーソル位置を1文字分戻す。
320	画面に空白を表示し、カーソル位置を1文字分戻す。
330	コメント
340	現在のカーソル位置を退避。
350	スクリーンエディットモード。
360	修正行1行をホストに送出。
370	カーソル位置を、スクリーンエディットモード以前に戻す。
390	コメント
410 } 420 } 430 } 440 } 450 }	エラーが発生した時、エラーコードを調べ、画面に表示し、エラー発生次のステートメントに戻る。
460	上記のエラーコード以外のエラー発生時、PC 8801のエラーメッセージを画面に表示。
470	エラー発生時次のステートメントに戻る。

```

490 'log-on routine
500 PRINT #1,"$$$CON,TSS",CR#;
510 PRINT "$$$CON,TSS"
520 RETURN
530 'terminal mode end routine
540 HELP OFF:KEY OFF:STOP OFF:CLOSE #1
550 ON ERROR GOTO 0
560 END
570 'send user-id and password routine
580 PRINT #1,"KADAIBANGO1$PASSWORD1";CR#;
590 PRINT "KADAIBANGO1"
600 RETURN
610 PRINT #1,"KADAIBANGO2$PASSWORD2";CR#;
620 PRINT "KADAIBANGO2"
630 RETURN
640 'send break routine
650 COLOR 2:PRINT "break";:COLOR 4
660 FOR W=1 TO 100:OUT &H21,&H3F:NEXT W:OUT &H21,&H37
670 RETURN

```

行番号	説明
490	コメント <span style="border: 1px solid black; padding: 2px;">f・1</span> キー押下により実行。
500	LOGON コマンドをホストに送信。この場合 JIS 端末としての LOGON になるので、英小文字を使用する時は、“\$\$\$ CON, TSS, ASC”に変更する。
530	コメント <span style="border: 1px solid black; padding: 2px;">f・10</span> キー押下により実行。
540	<span style="border: 1px solid black; padding: 2px;">HELP</span> キー、 <span style="border: 1px solid black; padding: 2px;">ファンクション</span> キー、 <span style="border: 1px solid black; padding: 2px;">STOP</span> キーをリセットし、RS 232C をクローズする。
550	エラー発生時の処理先をリセット。
570	コメント <span style="border: 1px solid black; padding: 2px;">f・2</span> キー押下により実行。
580	第1課題番号とパスワードをホストに送信。
590	第1課題番号を画面に表示。パスワードは表示しない。
610	第2課題番号とパスワードをホストに送信。 <span style="border: 1px solid black; padding: 2px;">f・3</span> キー押下により実行。
620	第2課題番号を画面に表示。パスワードは表示しない。
640	コメント <span style="border: 1px solid black; padding: 2px;">STOP</span> キー押下により実行。
650	画面に赤色で“break”を表示。
660	RS 232C にブレイク信号を送信。

```

680 'write file open routine
690 COLOR 2:INPUT "Enter write file name";F#:COLOR 4
700 IF F#="FILES" OR F#="files" THEN FILES:GOTO 690
710 OPEN F# FOR OUTPUT AS #2:LOGING=1
720 RETURN
730 'Stop the write file routine
740 COLOR 2:PRINT F#;" File close":COLOR 4
750 CLOSE #2:LOGING=0:RETURN
760 'Read file open routine
770 COLOR 2:INPUT "Enter read file name";F#:COLOR 4
780 IF F#="FILES" OR F#="files" THEN FILES:GOTO 770
790 OPEN F# FOR INPUT AS #2
800 LINE INPUT #2,L#
810 IF EOF(2) THEN 870
820 IF LOGING THEN 870
830 PRINT L#:PRINT #1,L#:LF#
840 IF LEFT$(L#,6)="$$$EOF" THEN 870
850 IF LISTING THEN LPRINT L#
860 GOTO 800
870 PRINT L#:PRINT #1,L#:LF#:CLOSE #2:LOGING=0:RETURN

```

行番号	説明
680	コメント <span style="border: 1px solid black; padding: 2px;">f・6</span> キー押下により実行。
690	受信データを書き込むフロッピーディスクファイル名の指定。
700	“FILES”を指定時は、フロッピーディスクのディレクトリーを表示。
710	フロッピーディスクオープン、フロッピーディスク書き込みスイッチオン。
730	コメント <span style="border: 1px solid black; padding: 2px;">f・7</span> キー押下により実行。
740	<span style="border: 1px solid black; padding: 2px;">f・6</span> キー押下により、オープンしたフロッピーディスクをクローズする。
760	コメント <span style="border: 1px solid black; padding: 2px;">f・8</span> キー押下により実行。
770	送信するデータが入っているフロッピーディスクファイル名の指定。
780	“FILES”を指定時は、フロッピーディスクのディレクトリーを表示。
790	フロッピーディスクオープン。
800	フロッピーディスクより1行入力。
810	入力データのチェック。
820	出力中止指令のチェック。
830	画面に表示すると共に、ホストに送信。
840	論理終了レコード“\$\$\$ EOF”のチェック。
850	ハードコピー指定があれば、プリント出力。
860	フロッピーディスクより1行入力へ戻る。
870	最終行を送信し、ファイルをクローズ。

```

880 'Help key routine
890 PRINT "f.1=logon      f.2=userid 1   f.3=userid 2 f.4=edit      f.5=list"
900 PRINT "f.6=file(w)   f.7=write-stop f.8=file(r)   f.9=read-stop  f.10=end"
910 RETURN
920 'Stop the read file routine
930 COLOR 2:PRINT F$;" File close":COLOR 4
940 LOGING=1:RETURN
950 'Listing key on routine
960 IF LISTING THEN LISTING=0:OUT &H10,&H13:OUT &H40,0:OUT &H40,1 ELSE LISTING=1
:OUT &H10,&H11:OUT &H40,0:OUT &H40,1:LPRINT CHR$(&H1B);"&"
970 RETURN

```

行番号	説明
880	コメント <span style="border: 1px solid black; padding: 2px;">HELP</span> キー押下により実行。
890	各ファンクションキーの機能を画面に表示。
900	
920	コメント <span style="border: 1px solid black; padding: 2px;">f・9</span> キー押下により実行。
940	フロッピーディスクからの送信を、強制終了させる出力中止スイッチオン。
950	コメント <span style="border: 1px solid black; padding: 2px;">f・5</span> キー押下により実行。
960	ハードコピー指定スイッチをオンにし、プリンターをONLINEにする。又すでにハードコピー指定スイッチがオンの時は、オフにし、プリンターをOFFLINEにする。

#### 4. 使用方法

PC8801 に第3章のプログラムを入力し、行番号580行の“KADAIBANGO1\$PASS - WORD1”と590行の“KADAIBANGO1”によく使用する課題番号とパスワードをセットする。同様に、610行と620行にも課題番号とパスワードをセットすることができる。

セットがすめば、このプログラムをフロッピーディスクか、カセットテープにSAVEしておくと、次回からはプログラムをロードするだけで済む。

ロードが終ればRUNコマンドでプログラムを実行する。これでPC8801はキャラクターディスプレイターミナルとして動作するので、電話でセンターを呼び出し、ピーと言うキャリー音を確認の上音響カブラーにセットすればよい。

f・1 キーを押下すると“\$\$\$CON, TSS”が送信され、ACOSとのLOGONが始まる。もしタイミングが悪くて

```
* N1020 14:52:33 NOT ACCEPT - NO CON
```

のメッセージが表示されたときは、再度 f・1 キーを押下すると

```
HANDA1 S1000 TSS ON 06/14/82 AT 14:52:55 CHANNEL 2704
```

```
USER ID -
```

と表示されるはずである。

課題番号はキーボードから入力してもよいが、このプログラムは2つの課題番号までは記憶させることができるので、その機能を利用するのが便利であろう。

**f・2**キー、又は、**f・3** キーで課題番号とパスワードが送信できる。これでLOGON は完了である。

いままで、端末を使ってこられた利用者は LOGONは **CTRL** キーと **A** キーを同時に押下するもの、と思ってこられた方もいるでしょうが、PC8801でこれをおこなうと HELP キーが押下された時と同じ結果になってしまうので注意が必要である。

バックスペースは **←** キーを押下、BREAKは **STOP** キーを押下することで出来るが、BASIC だけでプログラムを書いているためカーソル表示がないのが残念である。

あとは、普通のキャラクターディスプレイと同じである。もしPC8801のディスプレイにカラーディスプレイを使用されている方は、受信データは緑色、送信データは黄色と、見やすくなっているのに気がつくだろう。

次に、インテリジェント・ターミナルとしての機能説明を行う。

#### (1) スクリーンエディット機能

最近のパーソナル・コンピュータには、大型コンピュータ顔まけの強力なエディターを備えているものが多く、PC8801のスクリーンエディターも、その一つである。そこでこのスクリーンエディターを ONLINE 中に使用できる様にした。スクリーンエディターが使用できる時は、行番号つきのプログラムか、データに限られるが、少量の修正には非常に便利である。

まず、修正すべき所を画面に表示する。(スクリーンエディターなので画面に表示されていなければ修正できない。)つぎに **f・4** キーを押下するとカーソルが点滅するので修正すべき行に **←** **→** **↑** **↓** キーを用い、カーソルを移動させ、修正する。(削除、挿入は、PC8801 USER'S MANUAL 第9章スクリーンエディター参照) **≡** キーを押下すると修正した行、全体が送信されるので ACOS のカレントファイルを修正することができる。ただし、行の先頭に行番号がついていないものは修正できない。

#### (2) ハードコピー機能

キャラクターディスプレイ・ターミナルを使用していて、必要な時に、必要なだけのハードコピーが取れると便利だけでなく、用紙の節約にもなる。

**f・5** キーを押下すると、プリンターが ONLINE になり、以後、送受信はすべてプリント出力される。また、すでにハードコピーを取る様になっている時は、OFFLINE となり、以後送受信はプリント出力されなくなる。ただし、この機能はプリンターの種類によっては、使用中“Line Buffer overflow”が出力されるものもある。PC8023-C では正常に動作するがPC8821、

PC8822では30行ほど印字した所で overflow してしまうので注意を要する。

#### (3) 受信データのフロッピーディスクへの書き込み機能

受信したデータをフロッピーディスクへ記録できれば、後からそのデータをPC8801のBASICで処理したり、またプログラムであれば、サービス時間外にスクリーンエディター等でプログラムの修正もできるであろう。そのための機能がこれである。

**f·6** キーを押下すると、出力するフロッピーディスクのファイル名を問い合せてくるので、ファイル名を入力する。この時“FILES”又は“files”を入力するとフロッピーディスクのディレクトリー（ファイルの一覧表）が出力され、再びファイル名を問い合せてくる。

ファイル名を入力すると、そのファイルがオープンされ、以後受信データのみフロッピーディスクに書き込まれていく。フロッピーディスクのファイル名を入力した後、LISTコマンドでACOSファイルの内容を受信すれば、ACOSファイルの内容を記録することができるし、RUNコマンドを実行すればACOSで実行時に出力されるデータ（たとえば図形処理の座標データ）などを記録することができる。

受信が終れば、**f·7** キーを押下し、フロッピーディスクのファイルをクローズしておく。これを忘れると **f·7** キーが押されるまでフロッピーディスクに受信データが書き込まれることになる。

以下に実行例を付記する。

x Enter write file name ----- LIST-----         File close -----	----- <b>f·6</b> キー押下、フロッピーディスクのファイル名を入力。 -----カレント・ファイルの内容を出力。 ディスプレイに表示されると共に、フロッピーディスクに書き込まれる。 ----- <b>f·7</b> キー押下フロッピーディスクをクローズ。
---	---

#### (4) フロッピーディスクの転送機能

ACOSのTSSサービス時間外にPC8801のスクリーンエディターを使用し、プログラムやデータをフロッピーディスクに作成し、サービス時間にACOSファイルに転送できれば、電話代も節約になるし、仕事の能率が上がるであろう。そのための機能がこれである。

**f·8** キーを押下すると、転送すべきフロッピーディスクのファイル名を問い合せてくる。この時“FILES”、又は“files”を入力すると、フロッピーディスクのディレクトリーが出力され、再びファイル名を問い合せてくる。ファイル名を入力すると自動的にフロッピーディスクの内容が読み出されACOSに送信される。ファイルの内容が終れば、自動的にフロッピーディスクがクローズされる。

この機能を用いれば、ACOSのREADコマンドによってフロッピーディスクの内容を、ACOSのファイルに転送することができる。ただし、回線スピードが300BPSなので、1秒間に転送できる文字数は30文字弱である。

多量のデータを転送する時は、いくつかのブロックに分けて転送する方が、不慮のシステムダウン等の時、再度最初からのやり直しを防ぐことができる。

以下に実行例を付記する。

```

*
*READ PPT ----- READコマンドでPPTを指定する。(FDDでもよ
READY                                     い)
Enter read file name -- f·8 ----- f·8 キー押下により、フロッピーディスクのファイル
                                名を問い合せてくるので、ファイル名を入力すると、
                                ファイルの内容が送信される。
)
***EOF ----- フロッピーディスクの転送が終了すれば、ACOSに転送
*                                     の終了を示す$$$EOFを、キーボードから入力し、
                                     送信する。(フロッピーディスクの終りに $$$ EOF
                                     を記録しておけばキーボードから入力する必要はない。)

```

(5) その他

PC 8801 の **STOP** キーは、PC 8801 のBASICプログラムの中断キーである。もしターミナルプログラムを実行中に、過ってプログラムを中断すると困るので、このプログラムでは **STOP** キーをBREAKキーとして使用し、ターミナルプログラムを終了するには、 **f·10** キーを押下することにした。また、フロッピーディスクの内容をACOSファイルへ転送中に、中断したい時は、 **f·9** キーを押下することにより、随時中断できる様にしてある。

5. おまけ (PC 8801 での FORTRAN プログラム作成方法)

オフライン(PC 8801 単独で使用)で、FORTRAN プログラム又はデータを作成するには次の様にするとよい。(以下特に断わらないかぎり PC 8801 のコマンドを示す)

NEWコマンドでテキストエリアをクリアする。次に、AUTO コマンドで行番号を自動出力にする。ここで FORTRAN のプログラムを入力するのだが、このままでは、文番号が行番号と混同するので必ず始めに ' (アポストロフィー) を付け、コメント行としてプログラムを入力する。

入力が終われば、 **STOP** キーを押下し、アスキータイプでSAVEする(SAVE "ファイル名",A)とテキストエリアのプログラムがフロッピーディスクに記録されることになる。当然 AUTO コマンドでプログラム作成中は、スクリーンエディタが使用できるので修正、追加が素早くでき、カードパンチすることに比べたら、数倍便利である。フロッピーディスクにSAVEしてあるファイルも、LOAD コマンドでテキストエリアにLOADして、修正、追加が簡単にできる。

このフロッピーディスクにSAVEしてあるファイルを、ACOSに転送してやれば、FORTRANプログラムが出来上る。ただし' (アポストロフィー) が付いているので、ACOSのEDITサブシステムにてRS: / ' / ; \* : / / を実行して、' (アポストロフィー) を空白に置き換える必要がある。

## 6. 最 後 に

以上、PC8801をインテリジェント・ターミナルにするためのプログラムについて説明した。

PC8801のユーザマニュアルに、ターミナルとしての使用方法について記載されているが、実際ACOSと接続して使用する時に必要な情報は記載されていない、RS232Cをターミナルの送受信に使用する時のOPENコマンドが分らず苦労したが、一応ターミナルとして満足できるものが出来た。

今後の課題としては、1200BPS以上の回線スピードに追従できる様に、プログラムをBASICから機械語に変更することと、グラフィックディスプレイターミナルとしての機能を持つことなどが上げられる。

パーソナル・コンピューターは、プログラム次第で利用者の使用目的に合ったパーソナル・ターミナルに変身させることができる。この章がこの様な利用者の方々に少しでもお役に立てば幸いである。

### ( 参 考 文 献 )

- 1) 石田晴久「IF800をTSS端末として使うためのBASICプログラム」  
東京大学大型計算機センターニュース Vol.13, №4 30ページ
- 2) 高橋 理「IF800のACOS-TSSターミナル」  
東北大学大型計算機センターニュース Vol.14, №3 68ページ
- 3) 「PC-8801ユーザズマニュアル」(NEC)